# Copyright (c) 2011 Nokia Corporation.

QT += core

TARGET = MatchemPoker
TEMPLATE = app
VERSION = 1.1.0

INCLUDEPATH += spritebatch/src_general/ spritebatch/src_gles2spritebatch/

SOURCES += \
    src/tilesnpc.cpp \
    src/npclevel.cpp \
    src/TileInterfaces.cpp \
    src/ParticleEngine.cpp \
    src/QtTileGameEngine.cpp \
    src/GEInterfaces.cpp \
    src/GEAudioOut.cpp \
    src/GEAudioBuffer.cpp \
    src/GameWindow.cpp \
    src/main.cpp \
    spritebatch/src_general/SpriteBatch.cpp \
    spritebatch/src_gles2spritebatch/GLES2SpriteBatch.cpp


HEADERS  += \
    src/tilenpc.h \
    src/npclevel.h \
    src/TileInterfaces.h \
    src/ParticleEngine.h \
    src/QtTileGameEngine.h \
    src/GEInterfaces.h \
    src/GEAudioOut.h \
    src/GEAudioBuffer.h \
    src/GameWindow.h \
    spritebatch/src_general/SpriteBatch.h \
    spritebatch/src_gles2spritebatch/GLES2SpriteBatch.h


CONFIG += mobility
MOBILITY =


unix!symbian: {
    message(Unix build)

    QT += multimedia

    #only needed in unix desktop
    contains(DEFINES, DESKTOP) {
        INCLUDEPATH += ../SDKPackage_OGLES2/Builds/OGLES2/Include
        LIBS +=  -L../SDKPackage_OGLES2/Builds/OGLES2/LinuxPC/Lib

        INCLUDEPATH += $(HOME)/Downloads/qt-mobility-opensource-src-1.1.0/install/include
        INCLUDEPATH += $(HOME)/Downloads/qt-mobility-opensource-src-1.1.0/install/include/QtMultimedia
        LIBS +=  -L$(HOME)/Downloads/qt-mobility-opensource-src-1.1.0/install/lib

        message("The project contains desktop ogl")
        message($$INCLUDEPATH)
        message($$LIBS)
        LIBS += -lX11 -lEGL -lGLESv2
    }
}


maemo5: {
    message(Maemo build)

    QT += multimedia
    LIBS += -lX11 -lEGL -lGLESv2
}


windows: {
    message(Windows build)

    QT += multimedia
    INCLUDEPATH += /PowerVRSDK/Builds/OGLES2/Include
    LIBS +=  -L/PowerVRSDK/Builds/OGLES2/WindowsPC/Lib
    message("The project contains desktop ogl")
    message($$INCLUDEPATH)
    message($$LIBS)

    LIBS += -llibEGL -llibGLESv2
}


symbian: {
    message(Symbian build)

    # For orientation locking?
    # INCLUDEPATH += /Nokia/devices/Nokia_Symbian3_SDK_v0.9/epoc32/include/

    CONFIG += mobility
    MOBILITY += multimedia

    LIBS += -lcentralrepository
    LIBS += -llibEGL -llibGLESv2 -lcone -leikcore -lavkon

    message("The project contains desktop ogl")
    message($$INCLUDEPATH)
    message($$LIBS)
    message($$DEFINES)
    message($$QT)

    # FOR UGLY VOLUMEHACK
    INCLUDEPATH += /epoc32/include/mmf/common
    INCLUDEPATH += /epoc32/include/mmf/server
    # LIBS += -lmmfdevsound


    backup.sources = backup_registration.xml
    backup.path = !:/private/EE69dAED
    ICON = icon.svg

    TARGET.UID3 = 0xee69daed
    TARGET.EPOCSTACKSIZE = 0x14000
    TARGET.EPOCHEAPSIZE = 0x020000 0xf00000
}


RESOURCES += \
    textures.qrc \
    sounds.qrc


# Harmattan
unix:!symbian:!maemo5 {
    message(Harmattan build)
    DEFINES += Q_WS_HARMATTAN

    CONFIG += mobility
    MOBILITY += multimedia

    target.path = /opt/usr/bin

    desktopfile.files = qtc_packaging/debian_harmattan/matchempoker.desktop
    desktopfile.path = /usr/share/applications

    icon.files = matchempoker.png
    icon.path = /usr/share/icons/hicolor/64x64/apps

    INSTALLS += \
        target \
        desktopfile \
        icon
}




